pkgname='SPIRV-Cross'
_tag='2019-09-04'
pkgver=${_tag//-/.} # pkgver cannot contain hyphens.
_srcdir=$pkgname-$_tag
pkgrel=1
pkgdesc="A practical tool and library for performing reflection on SPIR-V and disassembling SPIR-V back to high level languages."
arch=('x86_64')
url="https://github.com/KhronosGroup/SPIRV-Cross"
license=('Apache-2.0')
optdepends=()
source=("https://github.com/KhronosGroup/$pkgname/archive/$_tag.tar.gz")
sha256sums=('b4e2e50c4514eda7dfbfeb3b80656a5306372e0ce904ba770c52fa903802d178')

build(){
  mkdir build
  cd build
  cmake \
    -DCMAKE_BUILD_TYPE=RelWithDebInfo \
    -DCMAKE_POSITION_INDEPENDENT_CODE=on \
    -DSPIRV_CROSS_ENABLE_TESTS=OFF \
    "../$_srcdir"
  make
}

package(){
  cd build
  make DESTDIR="$pkgdir" install
}

#check(){
#  cd build
#  make test
#}
# vim:ts=2:sw=2:et:
